[Efficient Excitation of Relativistic Ions for the Gamma Factory]

Simon Rochester
Rochester Scientific, LLC

Outline

Topics

- A brief introduction to the AtomicDensityMatrix package
- Laser excitation
    - Doppler-free cw
    - cw (with Doppler broadening)
    - Broadband cw
    - Adiabatic fast passage

Outline

Disclaimers

- The talk will be in the form of *Mathematica* code; however…
- …feel free to ignore code/equations. Useful information will be in the pictures!
- In the ADM package, $\hbar = c = 1$; I put them back in at the end to get numbers
- Various factors of 2 will sacrifice themselves in the name of nice-looking formulas
- Everything will be done in the ion CoM frame except where noted

AtomicDensityMatrix package

A brief introduction

- A *Mathematica* package for density-matrix-dynamics simulations in atomic and related systems
- Developed by Rochester Scientific (rochesterscientific.com)
- Distributed as open-source software (rochesterscientific.com/ADM)
- Can be used for numerical and analytical calculations
- Useful for educational models and full-scale simulations

Load the package:

In[363]:=
```
<< AtomicDensityMatrix`
```

The logo:

In[364]:=

```
ADMLogo[]
```

cw excitation

Basic idea

- Turn on light and pump faster than the upper state decays
- “cw” means a pulse longer than the decay time

cw excitation

Without Doppler broadening

- Define and solve the density-matrix evolution equations for excitation in a two-level system

Set some graphics options:

In[365]:=
```
ADMTheme["SlideShow"]
```

Define a two-level atomic system:

In[366]:=
```
sys = {
   AtomicState[1, Energy → 0, NaturalWidth → 0],
   AtomicState[2, BranchingRatio[1] → 1,
    NaturalWidth → Γ0, Energy → ω0]
  };
```

The Hamiltonian for the system subject to an optical field (Ω_R is Rabi frequency, ω is optical frequency):

In[367]:=
```
h = Hamiltonian[sys, ElectricField → OpticalField[ωL, ℰ]] /.
  ℰ → ΩR / ReducedME[1, {Dipole, 1}, 2]
```

$$\begin{pmatrix} 0 & -\cos(t\,\omega_L)\,\Omega_R \\ -\cos(t\,\omega_L)\,\Omega_R & \omega_0 \end{pmatrix}$$

- The Rabi frequency $\Omega_R = \frac{d\,\mathcal{E}}{\hbar}$ is the coupling strength for the transition

It depends on line width, transition frequency, and light intensity only

Rabi frequency in terms of light intensity $\mathcal{I}$ for later use:

In[368]:=

```
rabi = ΩR → d ℰ / ℏ /.
    {ℰ → Sqrt[2 ℐ / (c ϵ0)], d → Sqrt[ℏ c^3] Sqrt[4 π ϵ0] ExpandDipoleRME[
       sys, ReducedME[1, {Dipole, 1}, 2]]} // PowerExpand
```

$$\Omega_R \to \frac{c\sqrt{6\pi}\sqrt{\mathcal{I}}\sqrt{\Gamma_0}}{\sqrt{\hbar}\,\omega_0^{3/2}}$$

Apply the rotating-wave approximation to the Hamiltonian and write in terms of the detuning Δ from resonance:

In[369]:=

```
hrwa = RotatingWaveApproximation[sys, h, {ωL, 1 → 2, Δ}]
```

$$\begin{pmatrix} 0 & -\frac{\Omega_R}{2} \\ -\frac{\Omega_R}{2} & -\Delta \end{pmatrix}$$

Relaxation of the upper state due to spontaneous decay:

In[370]:=

```
relax = IntrinsicRelaxation[sys]
```

$$\begin{pmatrix} 0 & 0 \\ 0 & \Gamma_0 \end{pmatrix}$$

Repopulation of the lower state due to spontaneous decay:

In[371]:=

```
repop = OpticalRepopulation[sys]
```

$$\begin{pmatrix} \Gamma_0\,\rho_{2,2}(t) & 0 \\ 0 & 0 \end{pmatrix}$$

Here are the evolution equations:

In[372]:=
```
TableForm[eqs = LiouvilleEquation[sys, hrwa, relax, repop]]
```

$$\rho_{1,1}'(t) = \Gamma_0\, \rho_{2,2}(t) - i\left(\tfrac{1}{2}\, \Omega_R\, \rho_{1,2}(t) - \tfrac{1}{2}\, \Omega_R\, \rho_{2,1}(t)\right)$$
$$\rho_{1,2}'(t) = -\tfrac{1}{2}\, \Gamma_0\, \rho_{1,2}(t) - i\left(\tfrac{1}{2}\, \Omega_R\, \rho_{1,1}(t) + \Delta\, \rho_{1,2}(t) - \tfrac{1}{2}\, \Omega_R\, \rho_{2,2}(t)\right)$$
$$\rho_{2,1}'(t) = -\tfrac{1}{2}\, \Gamma_0\, \rho_{2,1}(t) - i\left(-\tfrac{1}{2}\, \Omega_R\, \rho_{1,1}(t) - \Delta\, \rho_{2,1}(t) + \tfrac{1}{2}\, \Omega_R\, \rho_{2,2}(t)\right)$$
$$\rho_{2,2}'(t) = -i\left(\tfrac{1}{2}\, \Omega_R\, \rho_{2,1}(t) - \tfrac{1}{2}\, \Omega_R\, \rho_{1,2}(t)\right) - \Gamma_0\, \rho_{2,2}(t)$$

Here are initial conditions, putting all the population in the ground state at $t = 0$:

In[373]:=
```
TableForm[
 inits = InitialConditions[sys, PopulatedDM[sys, 1], 0]]
```

$$\rho_{1,1}(0) = 1$$
$$\rho_{1,2}(0) = 0$$
$$\rho_{2,1}(0) = 0$$
$$\rho_{2,2}(0) = 0$$

Some assumptions on variables:

In[374]:=
```
$Assumptions = {ΩR > 0 ∧ Γ0 > 0 ∧ ΓD > 0 ∧ κ1 > 0 ∧
    Δ ∈ Reals ∧ ΔD ∈ Reals ∧ ΔL ∈ Reals ∧ ΓL > 0 ∧ r > 0};
```

Solve for resonant excitation:

In[375]:=
```
sols = First@
   DSolve[Join[eqs /. Δ → 0, inits], DMVariables[sys], t];
```

We can express the upper-state population in terms of a saturation parameter $\kappa_1 = \frac{\Omega_R^2}{{\Gamma_0}^2}$

Upper-state population in terms of κ_1:

In[376]:=

```
popK = ρ2,2[t] /. sols /. {ΩR → Γ0 √κ1} /. t → t / Γ0 // FullSimplify
```

$$\left(\exp\left(-\frac{1}{4}\,t\left(\sqrt{1-16\,\kappa_1}+3\right)\right)\kappa_1\right.$$
$$\left(2\exp\left(\frac{1}{4}\,t\left(\sqrt{1-16\,\kappa_1}+3\right)\right)(16\,\kappa_1-1)-\right.$$
$$\left(1+e^{\frac{1}{2}\,t\,\sqrt{1-16\,\kappa_1}}\right)(16\,\kappa_1-1)+3\,e^{\frac{1}{2}\,t\,\sqrt{1-16\,\kappa_1}}\,\sqrt{1-16\,\kappa_1}\,-$$
$$\left.\left.3\,\sqrt{1-16\,\kappa_1}\right)\right)\Big/(2\,(2\,\kappa_1+1)\,(16\,\kappa_1-1))$$

Plot the upper-state population as a function of time:

In[377]:=

```
SetOptions[Plot, PlotRange → {0, 1}];
SetOptions[Manipulate, SaveDefinitions → True];
With[{pop = popK},
 Manipulate[Plot[pop, {t, 0, 10.},
   FrameLabel → {"Time t/Γ0", "Upper-state population"}],
  {{κ1, 0, "κ1"}, 0, 100}]
]
```



As expected, after a time equal to the upper state lifetime, the population reaches a steady state

We can solve for the steady state directly

cw excitation

Steady state

To find steady state, set time derivatives to zero and use normalization $\mathrm{Tr}(\rho) = 1$

Evolution equations in the steady state:

In[380]:=

```
TableForm[steadyEqs =
  Append[Most[eqs /. ρ__′[t] → 0], Tr[DensityMatrix[sys]] == 1]]
```

$$0 = \Gamma_0\, \rho_{2,2}(t) - i\left(\tfrac{1}{2}\,\Omega_R\, \rho_{1,2}(t) - \tfrac{1}{2}\,\Omega_R\, \rho_{2,1}(t)\right)$$
$$0 = -\tfrac{1}{2}\,\Gamma_0\, \rho_{1,2}(t) - i\left(\tfrac{1}{2}\,\Omega_R\, \rho_{1,1}(t) + \Delta\, \rho_{1,2}(t) - \tfrac{1}{2}\,\Omega_R\, \rho_{2,2}(t)\right)$$
$$0 = -\tfrac{1}{2}\,\Gamma_0\, \rho_{2,1}(t) - i\left(-\tfrac{1}{2}\,\Omega_R\, \rho_{1,1}(t) - \Delta\, \rho_{2,1}(t) + \tfrac{1}{2}\,\Omega_R\, \rho_{2,2}(t)\right)$$
$$\rho_{1,1}(t) + \rho_{2,2}(t) = 1$$

Steady-state solution:

In[381]:=

```
TableForm[steadyStateSol =
  First@Solve[steadyEqs, DMVariables[sys]] // FullSimplify]
```

$$\rho_{1,1}(t) \to \frac{4\,\Delta^2+\Gamma_0^2+\Omega_R^2}{4\,\Delta^2+\Gamma_0^2+2\,\Omega_R^2}$$
$$\rho_{1,2}(t) \to -\frac{(2\,\Delta+i\,\Gamma_0)\,\Omega_R}{4\,\Delta^2+\Gamma_0^2+2\,\Omega_R^2}$$
$$\rho_{2,1}(t) \to \frac{(i\,\Gamma_0-2\,\Delta)\,\Omega_R}{4\,\Delta^2+\Gamma_0^2+2\,\Omega_R^2}$$
$$\rho_{2,2}(t) \to \frac{\Omega_R^2}{4\,\Delta^2+\Gamma_0^2+2\,\Omega_R^2}$$

Steady-state population of the upper state:

In[382]:=
```
steadyPop = ρ2,2[t] /. steadyStateSol /. ΩR^2 → ΩR^2 / 2
```

$$\frac{\Omega_R^2}{2\left(4\,\Delta^2+\Gamma_0^2+\Omega_R^2\right)}$$

- At low power, population is proportional to Ω_R^2 (light intensity)
- Width is given by Γ_0, but there is *power broadening* for $\kappa_1 > 1$
- At high power population saturates at 50%
- Again, the result can be written in terms of κ_1

Write population in terms of κ_1:

In[383]:=
```
steadyPopK =
 Cancel[steadyPop /. {ΩR^2 → κ1 Γ0^2, Δ → Δ Γ0}] /. Δ → Δ / Γ0
```

$$\frac{\kappa_1}{2\left(\frac{4\,\Delta^2}{\Gamma_0^2}+\kappa_1+1\right)}$$

Plot the upper-state population as a function of κ_1:

In[384]:=

```
With[{pop = steadyPopK /. {κ₁ → κ1, Γ₀ → 1}},
 Manipulate[Plot[pop, {κ1, 0, 10}, FrameLabel →
    {"Saturation parameter κ₁", "Upper-state population"}],
  {{Δ, 0, "Δ/Γ₀"}, -10, 10}]
]
```



Plot the upper-state population as a function of detuning:

In[385]:=

```
With[{pop = steadyPopK /. Γ₀ → 1},
 Manipulate[
  Plot[pop, {Δ, -10, 10}, FrameLabel → {"Detuning Δ/Γ₀",
     "Upper-state population"}], {κ₁, 0, 20}]
]
```



cw excitation

Numbers!

Numerical parameters for Li-like Pb, $2\,s_{1/2} \to 2\,p_{1/2}$:

In[386]:=

```
parameters = {
   ℏ → UnitConvert@Quantity["ReducedPlanckConstant"],
   c → UnitConvert@Quantity["SpeedOfLight"],
   ω0 -> UnitConvert[2 π
      Quantity[1 856 384., "SpeedOfLight" / "Centimeters"]],
   τ → Quantity[74. × 10^-12, "Seconds"],
   fractionalΔp -> 3. × 10^-4 ,
   γ → 96.3
  };
TableForm[parameters = Join[parameters,
   {Γ0 → 1 / τ, ΓD → fractionalΔp ω0} /. parameters]]
```

$\hbar \to 1.05 \times 10^{-34}$ kg m²/s

$c \to 299792458$ m/s

$\omega_0 \to 3.5 \times 10^{17}$ per second

$\tau \to 7.4 \times 10^{-11}$ s

fractionalΔp $\to 3. \times 10^{-4}$

$\gamma \to 96.3$

$\Gamma_0 \to 1.35 \times 10^{10}$ per second

$\Gamma_D \to 1.05 \times 10^{14}$ per second

Solve $\kappa_1 = 1$ for the saturation intensity:

In[388]:=

```
saturationI1 = First@Solve[ΩR^2/Γ0^2 == 1 /. rabi, I]
```

$$\left\{\mathcal{I} \to \frac{\hbar\,\Gamma_0\,\omega_0^3}{6\,c^2\,\pi}\right\}$$

Put in numerical parameters:

In[389]:=

```
saturationI1N = UnitConvert[
   I /. saturationI1 /. parameters, "Watts" / "Centimeters"^2]
```

3.6×10^6 W/cm^2

In the lab frame, the light intensity is scaled by the square of the Doppler factor $2\,\gamma$:

In[390]:=

```
saturationI1Lab = saturationI1N / (2 γ)^2 /. parameters
```

97. W/cm^2

cw excitation

With Doppler broadening

- Longitudinal momentum spread leads to a CoM Doppler width $\Gamma_D = \frac{\Delta p}{p}\,\omega_0$

- Need to weight population according to Doppler distribution and integrate over Doppler shift

A Gaussian Doppler distribution:

In[391]:=

```
dopplerDistribution = Exp[-(ΔD / ΓD)^2] / (ΓD Sqrt[π])
```

$$\frac{e^{-\frac{\Delta_D^2}{\Gamma_D^2}}}{\sqrt{\pi}\,\Gamma_D}$$

Plot the distribution:

In[392]:=

```
Plot[dopplerDistribution /. ΓD → 1,
 {ΔD, -3, 3}, FrameLabel → {"Doppler shift ΔD/ΓD"},
 Filling → Bottom, PlotRange → All]
```



Doppler-weighted upper-state populations, assuming $\Omega_R \gg \Gamma_0$:

In[393]:=

```
steadyPopDoppler =
 dopplerDistribution (steadyPop /. {Δ → ΔD, Γ0 → 0})
```

$$\frac{e^{-\frac{\Delta_D^2}{\Gamma_D^2}}\,\Omega_R^2}{2\sqrt{\pi}\,\Gamma_D\left(4\,\Delta_D^2+\Omega_R^2\right)}$$

Integrate over Doppler shift:

In[394]:=
```
steadyPopIntegrated = ∫_{-∞}^{∞} steadyPopDoppler ⅆΔD
```

$$\frac{e^{\frac{\Omega_R^2}{4\Gamma_D^2}}\sqrt{\pi}\,\mathrm{erfc}\left(\frac{\Omega_R}{2\Gamma_D}\right)\Omega_R}{4\,\Gamma_D}$$

The population in this case can be written in terms of a different saturation parameter $\kappa_2 = \frac{\Omega_R^2}{\Gamma_D^2}$

Write in terms of κ_2:

In[395]:=
```
steadyPopIntegratedK =
  steadyPopIntegrated /. {ΩR → 2 √κ2 ΓD} // FullSimplify
```

$$\frac{1}{2} e^{\kappa_2}\sqrt{\pi}\,\mathrm{erfc}\left(\sqrt{\kappa_2}\right)\sqrt{\kappa_2}$$

Plot population as a function of κ_2:

In[396]:=

```
Plot[steadyPopIntegratedK, {κ2, 0, 5}, FrameLabel →
  {"Saturation parameter κ2", "Upper-state population"}]
```



Thus we need $\kappa_2 = \frac{\Omega_R^2}{\Gamma_D^2} \sim 1$ to obtain 50% excitation

cw excitation

Numbers!

- Saturation intensity for κ_2 is larger than for κ_1 by a factor $\frac{\Gamma_D^2}{\Gamma_o^2}$

The scaling factor:

In[397]:=
```
scaleFactor12 = ΓD^2/Γ0^2 /. parameters
```

6.03×10^7

Saturation intensity for Doppler-broadened cw pumping:

In[398]:=
```
saturationI2N = saturationI1N scaleFactor12
```

2.17×10^{14} W/cm^2

In the lab frame:

In[399]:=
```
saturationI2Lab = saturationI1Lab scaleFactor12
```

5.84×10^9 W/cm^2

- Much worse!

cw excitation

With a broadband laser

- We can do better by broadening the laser profile to match the Doppler width

- Divide up the Doppler distribution into segments of width $\sim\Gamma_o$ and apply light resonant with each segment

Illustrate the segmentation of the Doppler distribution:

In[400]:=

```
Show[Plot[dopplerDistribution /. ΓD → 1,
   {ΔD, -3, 3}, Filling → Bottom], ListPlot[
   Table[{ΔD, dopplerDistribution /. ΓD → 1}, {ΔD, -3, 3, .25}],
   PlotStyle → None, Filling → Bottom, FillingStyle → Black],
  PlotRange → All, FrameLabel → {"Doppler shift ΔD/ΓD"}]
```



- We want to just saturate each segment, i.e., $\kappa_1 = 1$ for the light resonant with each segment, or $(\Omega_R^2)_{\text{segment}} = \Gamma_0^2$

- Ω_R^2 is proportional to intensity, and intensities add

- With $\sim \frac{\Gamma_D}{\Gamma_0}$ segments, $(\Omega_R^2)_{\text{total}} = \frac{\Gamma_D}{\Gamma_0} (\Omega_R^2)_{\text{segment}} = \Gamma_D \Gamma_0$

- Thus the new saturation parameter is $\kappa_3 = \frac{\Omega_R^2}{\Gamma_D \Gamma_0}$

cw excitation

More numbers!

- Saturation intensity for κ_3 is smaller than for κ_2 by a factor $\frac{\Gamma_D}{\Gamma_o}$

The scale factor:

In[401]:=
```
scaleFactor23 = ΓD/Γ0 /. parameters
```

7760.

Saturation intensity for AFP:

In[402]:=
```
saturationI3N = saturationI2N / scaleFactor23
```

2.79×10^{10} W/cm^2

In the lab frame:

In[403]:=
```
saturationI3Lab = saturationI2Lab / scaleFactor23
```

7.53×10^{5} W/cm^2

- Better!

Adiabatic fast passage

Basic idea

- Another technique is to sweep the light frequency through resonance
- This adiabatically swaps the lower and upper states, transferring all atoms from one to the other
- We can show this by diagonalizing the Hamiltonian

Hamiltonian in the rotating-wave approximation:

In[404]:=
```
hrwa
```

$$\begin{pmatrix} 0 & -\frac{\Omega_R}{2} \\ -\frac{\Omega_R}{2} & -\Delta \end{pmatrix}$$

- Under the RWA, which state is “upper” depends on the light detuning

Give the Hamiltonian a symmetric appearance by shifting the reference energy:

In[405]:=
```
symmetricH = hrwa + Δ / 2 IdentityMatrix[2]
```

$$\begin{pmatrix} \frac{\Delta}{2} & -\frac{\Omega_R}{2} \\ -\frac{\Omega_R}{2} & -\frac{\Delta}{2} \end{pmatrix}$$

This Hamiltonian shows the symmetry between the two states under $\Delta \to -\Delta$

Find the eigenenergies and eigenstates:

In[406]:=

```
{eigenenergies, eigenstates} = Eigensystem[symmetricH]
```

$$\begin{pmatrix} -\frac{1}{2}\sqrt{\Delta^2+\Omega_R^2} & \frac{1}{2}\sqrt{\Delta^2+\Omega_R^2} \\ \left\{-\frac{\Delta-\sqrt{\Delta^2+\Omega_R^2}}{\Omega_R}, 1\right\} & \left\{-\frac{\Delta+\sqrt{\Delta^2+\Omega_R^2}}{\Omega_R}, 1\right\} \end{pmatrix}$$

Normalize eigenstates:

In[407]:=

```
eigenstates = Normalize /@ eigenstates // FullSimplify
```

$$\begin{pmatrix} \frac{1}{\sqrt{\frac{2\Delta\left(\Delta+\sqrt{\Delta^2+\Omega_R^2}\right)}{\Omega_R^2}+2}} & \frac{1}{\sqrt{\frac{2\Delta\left(\Delta-\sqrt{\Delta^2+\Omega_R^2}\right)}{\Omega_R^2}+2}} \\ -\frac{1}{\sqrt{\frac{2\Delta\left(\Delta-\sqrt{\Delta^2+\Omega_R^2}\right)}{\Omega_R^2}+2}} & \frac{1}{\sqrt{\frac{2\Delta\left(\Delta+\sqrt{\Delta^2+\Omega_R^2}\right)}{\Omega_R^2}+2}} \end{pmatrix}$$

Plot them:

In[408]:=

```
colorFunc[state_] = Blend[{Red, Blue}, state^2 /. Δ → #] &;
With[{energies = eigenenergies /. ΩR → ΩR,
  states = First@eigenstates /. ΩR → ΩR}, Manipulate[Column[{
    Show@MapThread[Plot[#1, {Δ, -10, 10},
        FrameLabel → {"", "Eigenenenergies"}, PlotRange → 5,
        ImagePadding → {{60, 2}, {30, 10}}, ColorFunction →
         colorFunc[#2], ColorFunctionScaling →
         False] &, {energies, states}],
    Plot[states, {Δ, -10, 10}, FrameLabel →
      {"Detuning Δ", "Eigenstate amplitudes"},
     ImagePadding → {{60, 2}, {60, 10}}]}, Spacings → 0],
  {{ΩR, 0.001, "ΩR"}, 0.001, 5}]
]
```



- There is an avoided crossing with splitting Ω_R

- At the crossing, the eigenstates swap their identification with the lower and upper states

- If we do a slow enough sweep, atoms will stay in their eigenstates without noticing that the eigenstates have swapped positions

Adiabatic fast passage

Let’s try it out!

The evolution equations with a linear sweep of the detuning at rate r and central detuning Δ_0:

In[410]:=

```
TableForm[chirpEqs = eqs /. {Δ → Δ0 + r t, Γ0 → 0}]
```

$$\rho_{1,1}'(t) = -i\left(\frac{1}{2}\,\Omega_R\,\rho_{1,2}(t) - \frac{1}{2}\,\Omega_R\,\rho_{2,1}(t)\right)$$
$$\rho_{1,2}'(t) = -i\left(\frac{1}{2}\,\Omega_R\,\rho_{1,1}(t) + (r\,t + \Delta_0)\,\rho_{1,2}(t) - \frac{1}{2}\,\Omega_R\,\rho_{2,2}(t)\right)$$
$$\rho_{2,1}'(t) = -i\left(-\frac{1}{2}\,\Omega_R\,\rho_{1,1}(t) - (r\,t + \Delta_0)\,\rho_{2,1}(t) + \frac{1}{2}\,\Omega_R\,\rho_{2,2}(t)\right)$$
$$\rho_{2,2}'(t) = -i\left(\frac{1}{2}\,\Omega_R\,\rho_{2,1}(t) - \frac{1}{2}\,\Omega_R\,\rho_{1,2}(t)\right)$$

Put the population in the ground state at $t = -10$:

In[411]:=

```
TableForm[
 inits = InitialConditions[sys, PopulatedDM[sys, 1], -10]]
```

$$\rho_{1,1}(-10) = 1$$
$$\rho_{1,2}(-10) = 0$$
$$\rho_{2,1}(-10) = 0$$
$$\rho_{2,2}(-10) = 0$$

Find the solution numerically:

In[412]:=

```
sol = ParametricNDSolveValue[
  {chirpEqs, inits}, ρ2,2, {t, -10, 10}, {ΩR, Δ0, r}]
```

ParametricFunction[Expression: $\rho_{2,2}$ Parameters: $\{\Omega_R, \Delta_0, r\}$]

Plot the populations as a function of time:

In[413]:=

```
Manipulate[pop = sol[ΩR, Δ0, r];
 Plot[pop[t], {t, -10, 10},
  FrameLabel → {"Time t", "Upper-state population"}],
 {{ΩR, 0, "ΩR"}, 0, 10}, {r, -10, 10}, {{Δ0, 0, "Δ0"}, -10, 10}]
```



- For high enough power and slow enough sweep rate, essentially all of the atoms end up in the upper state
- As long as the conditions are met, it is quite insensitive to the exact value of power, sweep rate, and Doppler shift

Adiabatic fast passage

Adiabatic condition

- Need to avoid transitions between eigenstates
- Condition is (“rotational speed of eigenstates”)$_{max}$ $\ll$ (transition frequency)$_{min}$
- Rotational speed of eigenstates is $\left|\left\langle \psi_1 \middle| \frac{d}{dt} \middle| \psi_2 \right\rangle\right|$

Find maximum rotational speed of eigenstates:

In[414]:=
```
{es1, es2} = eigenstates /. Δ → r t;
maxRotation = Abs[es1.D[es2, t]] /. t → 0 // FullSimplify
```

$$\frac{r}{2\,\Omega_R}$$

Find minimum transition frequency:

In[416]:=
```
minFrequency =
 First@Differences@eigenenergies /. Δ → 0 // FullSimplify
```

$$\Omega_R$$

Find the adiabatic parameter:

In[417]:=
```
minFrequency
------------
maxRotation
```

$$\frac{2\,\Omega_R^2}{r}$$

Thus we require $\frac{\Omega_R^2}{r} \gg 1$

$r = \frac{\text{chirp frequency range}}{\text{chirp time}} \sim \frac{\Gamma_D}{\tau} = \Gamma_D \Gamma_0$

So saturation parameter is the same as for broadband light: $\kappa_3 = \frac{\Omega_R^2}{\Gamma_D \Gamma_0} \gg 1$

Conclusion

Summary

- Different excitation techniques require different saturation intensities

Saturation intensities:

In[418]:=
```
Grid[{{"Doppler-free cw", saturationI1Lab},
  {"cw", saturationI2Lab},
  {"Broadband cw", saturationI3Lab},
  {"AFP", saturationI3Lab}}, Frame → True]
```

Doppler−free cw	$97.\ \mathrm{W/cm^2}$
cw	$5.84\times10^{9}\ \mathrm{W/cm^2}$
Broadband cw	$7.53\times10^{5}\ \mathrm{W/cm^2}$
AFP	$7.53\times10^{5}\ \mathrm{W/cm^2}$

- AFP and broadband cw techniques have the same power requirements, but AFP could provide twice as many excited ions
- AFP could be implemented without needing to chirp the laser by using a diverging laser beam, so that ions experience a changing Doppler shift as they travel through the beam